

## **Electronic Supplementary Information (ESI)**

### ***Synergy and sensitivity-balance in concatenating experiments in NO Relaxation Delay NMR (NORD)***

*István Timári, Tamás Milán Nagy, Katalin E. Kövér, Ole W. Sørensen*

## Table of Contents

1. NMR acronyms
2. Figure S1. One- and two-module truncated versions of the NORD {HMBC}-{HSQC}-{TOCSY} experiment
3. Figure S2. Optimization of single-module NORD experiments
4. Figure S3. Excerpts from the three-module NORD {HMBC}-{HSQC}-{TOCSY} experiment on a monosaccharide derivative
5. Figure S4. Sensitivity comparison of individual NORD {HMBC} versus HMBC from NORD {HMBC}-{HSQC}-{TOCSY} three-module experiment on a monosaccharide derivative
6. Figure S5. Sensitivity comparison of individual NORD {HMBC} versus HMBC from NORD {HMBC}-{HSQC}-{TOCSY} three-module experiment on a monosaccharide
7. Figure S6. Sensitivity comparison of individual NORD {HSQC} versus HSQC from NORD {HSQC}-{TOCSY} two-module experiment
8. Figure S7. Spin-system assignment of a decapeptide with time-shared two-step multiplicity edited NORD {HSQC}-{TOCSY}
9. Pulse sequence codes for Bruker spectrometers

## **NMR acronyms**

BANGO: small angle or  $\beta$ -TANGO

BIRD: Bilinear Rotation Decoupling

TIG-BIRD: Triselective Independent Gyration BIRD

BIG-BIRD: Biselective Independent Gyration BIRD

HMBC: Heteronuclear Multiple Bond Correlation

HSQC: Heteronuclear Single Quantum Coherence


INEPT: Insensitive Nuclei Enhancement by Polarization Transfer

NORD: NO Relaxation Delay


TOCSY: Total Correlation Spectroscopy

TANGO: Testing for Adjacent Nuclei with a Gyration Operator


**NORD {HMBC}**




**NORD {TOCSY}**




**Standard NORD {HSQC}**




**Updown NORD {HSQC}**




**Standard NORD {HSQC}-{TOCSY}**




**Updown NORD {HSQC}-{TOCSY}**




**Standard NORD {HMBC}-{HSQC}**




**Updown NORD {HMBC}-{HSQC}**




### NORD {HMBC}-{TOCSY}



**Figure S1** One- and two-module truncated versions of the NORD {HMBC}-{HSQC}-{TOCSY} experiment outlined in Fig. 1. See the caption to that figure for details.




**Figure S2.** Optimization of single-module NORD experiments on a monosaccharide (34 mg/550 $\mu\text{l}$   $\text{CDCl}_3$ ). Firstly,  $^1\text{H}$  excitation angle ( $\beta$ ) dependence test measurements in the NORD {HMBC} (A) were carried out. The number of scans was 2 in the 9 minutes long experiments and the excitation flip angle was incremented in  $10^\circ$  steps from  $90^\circ$  to  $170^\circ$ . Prepolarization time (DIPS1-2) dependence in the NORD {HSQC} experiment (B) was also measured. The number of scans was 2 in the 7 minutes long experiments and the DIPS1-2 time was incremented in 10 ms steps between 10 ms and 80 ms.  $^1\text{H}$  excitation angle ( $\alpha$ ) dependence in the NORD {TOCSY} (C) was also investigated. The number of scans was 2 in the 10.5 minutes long experiments and the excitation angle was incremented in  $10^\circ$  steps from  $40^\circ$  to  $70^\circ$ . The direct acquisition time ( $t_{2_{\text{max}}}$ ) was 143.4 ms.



a)


b)

c)




**Figure S3.** Excerpts from the NORD {HMBC}-{HSQC}-{TOCSY} experiment on a monosaccharide derivative (**1**, 60 mg dissolved in 550 $\mu$ l DMSO- $d_6$ /C $_6$ D $_6$ ). The measurement time was 26.5 minutes, two scans were recorded per increment, with 1024 points acquired in  $t_1$  and  $t_2$  in each module. The spectral width was 5.1 ppm in the  $^1\text{H}$  and 100 ppm in the  $^{13}\text{C}$  dimension. The direct acquisition time ( $t_{2\text{max}}$ ) was 143.4 ms. The TOCSY mixing time was 80 ms and the excitation angle  $\alpha$  was set to  $30^\circ$ . Adiabatic  $^{13}\text{C}$  inversion pulses were applied for better sensitivity: CAWURST-20 (121 ppm, 1.94 ms; H2L) and CAWURST-20 (121 ppm, 0.97 ms; H2L). Spectral data were extracted and presented in three correlation maps (a-c): HMBC, HSQC and TOCSY, respectively.






**Figure S4.** Sensitivity comparison of individual NORD {HMBC} (black) versus HMBC from NORD {HMBC}-{HSQC}-{TOCSY} three-module experiments (red). The measurements were carried out on a monosaccharide sample (**1**), 60 mg dissolved in 550  $\mu\text{L}$   $\text{DMSO-}d_6/\text{C}_6\text{D}_6$ . After 26.5 minutes runtime, 1024 increments were collected ( $t_1$ ) with 1024 data points ( $t_2$ ). The number of scans was 2 and the spectral width was set to 5.1 ppm ( $^1\text{H}$ ) and 100 ppm ( $^{13}\text{C}$ ). The direct acquisition time ( $t_{2_{\text{max}}}$ ) was 143.4


ms. To reach the optimal sensitivity balance, magnetization was excited with  $\beta^l = \frac{\pi}{2}$  in BANGO of the HMBC module and  $\alpha = 30^\circ$  in the TOCSY module of NORD {HMBC}-{HSQC}-{TOCSY}. On the other hand, a  $130^\circ$   $^1\text{H}$  excitation pulse was used in the corresponding standalone NORD {HMBC}. A range of 1.7-2.4 S/N improvement was observed in the concatenated HMBC shown with five different sections relative to the corresponding individual optimized NORD {HMBC} experiment with the same number of scans. Red spectra are horizontally shifted for better visualization and the measured S/N values are given above the  $F_2$  sections.



**Figure S5.** Sensitivity comparison of individual NORD {HMBC} (black) versus HMBC from NORD {HMBC}-{HSQC}-{TOCSY} experiment (red). Measurements were carried out on a monosaccharide sample (structure shown on the top), 30 mg dissolved in 550  $\mu\text{L}$   $\text{DMSO-d}_6$  solution. In 26.5 minutes runtime, 1024  $t_1$  increments were collected with 1024  $t_2$  data points. The number of scans was 2 and the spectral widths were set to 5.1 ppm ( $^1\text{H}$ ) and 100 ppm ( $^{13}\text{C}$ ). The direct acquisition time ( $t_{2\text{max}}$ ) was 143.4 ms. An average of 2.0 S/N improvement was observed in the HMBC spectrum from NORD {HMBC}-{HSQC}-{TOCSY} relative to the corresponding individual optimized NORD {HMBC} experiment with the same number of scans. Red spectra are horizontally shifted for better visualization.



**Figure S6.** Sensitivity comparison of NORD {HSQC} (black) compared to the HSQC spectrum from NORD {HSQC}-{TOCSY} (red). S/N values measured in NORD {HSQC}-{TOCSY} are given above the  $F_2$  sections relative to the corresponding individual optimized NORD {HSQC} experiment with the same number of scans. Red signals are shifted horizontally for better visualization. The demonstrated correlations were selected by their coupling patterns: the singlet Hb, H1 with a small homonuclear coupling (3.6 Hz) and H4 with a large coupling (9.4 Hz). The two-module experiment was recorded in 17 minutes on compound **1** (60 mg/ 550  $\mu$ l DMSO- $d_6$ /C $_6$ D $_6$ ). 2 scans were recorded per increment, with 1024 points acquired in  $t_1$  and  $t_2$  in both HSQC and TOCSY modules. The spectral width was 5.1 ppm in the  $^1\text{H}$  and 100 ppm in the  $^{13}\text{C}$  dimension. The direct acquisition time ( $t_{2\text{max}}$ ) was 143.4 ms. The TOCSY mixing time was 80 ms and the  $\alpha$  angle was  $30^\circ$ . The stand-alone NORD {HSQC} was acquired in 8 minutes with identical experimental parameters.



**Figure S7.** Spin-system assignment of a decapeptide with ten different residues (ADFIKLSTVG) using the edited NORD {HSQC}-{TOCSY} experiment. The sample contained 10.16 mg of peptide in 550  $\mu\text{l}$  DMSO- $d_6$  solution. The plots shown are the overlay of CH/CH $_3$  (black) and CH $_2$  (red)  $^{13}\text{C}$ - $^1\text{H}$  HSQC subspectra in (A) with expansion of the  $\text{C}\alpha$ - $\text{H}\alpha$  region and  $^1\text{H}$ - $^1\text{H}$  TOCSY (B) with expansion of the H-NH region to the left. The concatenated experiments were run in 24 minutes and 2 scans were recorded per increment (512 in  $t_1$ ) in a time-shared two-step editing cycle. The number of  $t_2$  data points was 2048 in both modules. The direct acquisition time ( $t_{2\text{max}}$ ) was 174.1 ms. The TOCSY mixing time was 80 ms.

**Pulse sequence code of NORD {HMBC}-{HSQC}-{TOCSY} experiment for Bruker spectrometers  
(Avance III and NEO systems)**

```
;nord_hmhc_hsqc_tocsy
;avance-version - tested on NEO system, TopSpin 4.0.2

;This pulse sequence is part of István Timári, Tamás Milán Nagy, Katalin E. Kövér,
;Ole W. Sørensen, 'Synergy and sensitivity-balance in concatenating experiments
;in NO Relaxation Delay NMR (NORD)'

;June 17, 2021 tested by IT, TMN, KEK

;The pulse sequence has been coded for test purposes only and may contain errors.
;The functionality of the pulse sequence itself may differ depending on
;the hardware as well as the software used to execute it. Functionality
;on differing systems cannot be granted.
;Any use of this pulse sequence on a spectrometer is at your own risk.
;
;By using this pulse sequence, or any modification of it in any published material
;you agree to acknowledge the above-mentioned publication.

;NORD concatenation of THREE experiments: HMBC, HSQC and TOCSY

;BANGO, TIG-BIRD and INEPT elements include J-compensated adiabatic X-inversion pulse

;set FNmode - echo-antiecho and NBL=3

;PROCESSING:
;FIRST separate the three datasets using AU splitx
;the resulting datasets can be processed with xfb (set echo-antiecho, pk)
;then for hmhc use xf2m for magnitude calculation in F2
;the F1 chemical shift scale in TOCSY is corrected with au-program fixF1 prior to xfb.

;CLASS=HighRes
;DIM=2D
;TYPE=
;SUBTYPE=
;COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delayek.incl>

;use: wvm -a in TopSpin command line to setup adiabatic pulse parameters

"cnst2=113"           ;for J-compensated adiabatic pulse set CNST2 = 113

"cnst31=(1-sfo2/sfo1)/(1+sfo2/sfo1)"

define list<gradient> EA3 = { 1.000 -cnst31}   ;HMBC phase sensitive, echo-anti echo
define list<gradient> EA4 = { -cnst31 1.000}

"p2=p1*2"
"p4=p3*2"

"d4=1s/(cnst2*4)"    ;INEPT delay
"d6=1s/(cnst2*2)"    ;BANGO and TIG-BIRD delays

"d11=50u"

"d20=10u"           ;z-filter delays in TOCSY
"d21=10u"

"p11=p1/2"          ;1H-45 for TIG-BIRD with v = 0

"p13 = (p1/90)*cnst11" ;cnst11 = 45 corresponds to beta = 90 in BANGO
```

```

"d0=3u"
"in0=inf1/2"

"d10=3u"

"cnst10=2*dw/inf1" ;cnst10 is a scaling factor for SW(13C)/SW(1H) for TOCSY
"in10=inf1*cnst10"

"DELTA=p16+d16+p2+d0*2-4u"

"DELTA2=d4-4u" ;INEPT
"d22 = DELTA2-p29"

"d25=p16+d16+4u-p1*0.78+de"

"d24 = 1.86m"

"DELTA3=d24-cnst17*p24/2-4u" ;d24 = 1.86 ms, CNST17 = 1

"DELTA4=d6-4u" ;BANGO, TIG-BIRD,
"d23 = DELTA4-p31"

"DELTA5=p16+d16+d10" ;1H-1H TOCSY
"DELTA6=p16+d16+8u"

"d18=1s/(cnst14*2)" ;cnst14 = 6 for HMBC long-range J(CH)

"DELTA21=1s/(2 * cnst6)-p16-d16" ;HMBC
"DELTA22=1s/(2 * cnst7)-p16-d16" ;CNST6 = 125 Hz, CNST7 = 165 Hz
"DELTA23=d18-p16-d16-4u"
"DELTA24=p2+d0*2"

"FACTOR3=(d7/(p6*115.112))/2" ;d7 = 70-80 ms for 1H-1H TOCSY transfer
"l3=FACTOR3*2"

"l4 = td1/6" ;NBL = 3, data from three experiments are stored in one file

"acqt0=0"

"p12 = (p1/90)*cnst12" ;CNST12 = small flip angle excitation pulse of 1H-1H TOCSY
"l9=1"

baseopt_echo

1 ze
  d11 p112:f2 ;d11 = 50u
2 d11 do:f2 st0 ;NBL = 3
  d11 p12:f2 p11:f1

(p3 ph1):f2
50u UNBLKGRAD
p16:gp12*2.77 ;purging gradient, gpz12 = 17.13%
d16

3 (p13 ph1):f1 ;HMBC starts with BANGO resulting in 90 degree excitation
;of 1H-12C proton magn., CNST11 sets BANGO angle
DELTA4 p10:f2
4u

(p2 ph1)
(p31:sp18 ph1):f2 ;J-compensated adiabatic pulse during BANGO, H2L sweep, 1.94 ms

4u
d23 p12:f2
(p13 ph1):f1

```

```

DELTA21                ;second order low-pass J-filter to suppress one-bond correlations
p16:gp23                ;gpz23 = 15
d16 p12:f2
(p3 ph27):f2
DELTA22
p16:gp14                ;gpz14 = -10
d16
(p3 ph27):f2
4u
p16:gp15                ;gpz15 = -5
d16
DELTA23
(p3 ph28):f2
d0
(p2 ph26):f1
d0
p16:gp1*EA3            ;gpz1 = 80
d16
(p24:sp7 ph29):f2      ;p24 = 2msec for Crp80comp.4
DELTA24
p16:gp1*EA4
d16 p12:f2
(p3 ph29):f2
4u BLKGRAD

goscnp ph15

12 d11 st
13 d11 do:f2

50u UNBLKGRAD

p16:gp13                ;purging gradient between two experiments
                        ;gpz13 = 17.13-71%
d16 p11:f1
                        ;HSQC starts with TIG-BIRD

14 (p11 ph11):f1        ;TIG-BIRD with v = 0, p11=45 degree 1H; ph11=phase 45
DELTA4 p10:f2
4u

(p2 ph1)
(p31:sp18 ph6):f2      ;J-compensated adiabatic pulse, H2L sweep, 1.94 ms
4u

d23
(p1 ph12):f1           ;ph12 = phase 225 degree

DELTA2                  ;INEPT delay
4u

(p2 ph1)
(p29:sp20 ph6):f2      ;J-compensated adiabatic pulse, H2L sweep, 0.97ms
4u

d22 p12:f2

(p1 ph22)
p16:gp11                ;purging gradient gpz11 = 31-51 during zz-magn.
d16

(p3 ph3):f2
d0

(p2 ph7)

d0
p16:gp1*EA
d16 p10:f2

```

```

(p24:sp7 ph8:r):f2      ;composite CHIRP, 2 ms
4u
DELTA p12:f2

(center (p1 ph21) (p3 ph4):f2 )
4u
DELTA3 p10:f2
(center (p2 ph1) (p24:sp7 ph9:r):f2 )      ;composite CHIRP, 2 ms
4u
DELTA3 p12:f2
(center (p1 ph2) (p3 ph5):f2 )
4u
DELTA2 p10:f2

(p2 ph1)
(p29:sp20 ph1):f2      ;J-compensated adiabatic pulse, H2L sweep, 0.97ms
4u
d22 p11:f1

(p1 ph1)

d25

(p2 ph1)

4u p112:f2
p16:gp2                ;gpz2 = 20.1
d16 BLKGRAD

goscnp ph31 cpd2:f2

22 d11 st
23 d11 do:f2

50u UNBLKGRAD
p16:gp0                ;purging gradient between experiments
                        ;gpz0 = 17.13-71%

d16 p11:f1

p12 ph16      ;1H-1H TOCSY with DIPSI-2 starts using small-flip angle excitation
if "19 %2 == 1"      ;TOCSY echo-antiecho selection

{
DELTA5      ;p12=(p1/90)*cnst12", CNST12=small flip angle excitation pulse
p2 ph17
d10          ;d10 incremented by in10 to get 1H-dimension for TOCSY in F1
p16:gp3      ;gpz3 = 30
d16
}
else
{
d10
p16:gp3*EA
d16
p2 ph17
DELTA5
}

p1 ph18
d20 p110:f1

                        ;begin DIPSI2
24 p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23

```



p6\*2.722 ph25  
p6\*4.167 ph23  
p6\*2.944 ph25  
p6\*4.111 ph23

p6\*3.556 ph25  
p6\*4.556 ph23  
p6\*3.222 ph25  
p6\*3.167 ph23  
p6\*0.333 ph25  
p6\*2.722 ph23  
p6\*4.167 ph25  
p6\*2.944 ph23  
p6\*4.111 ph25

p6\*3.556 ph25  
p6\*4.556 ph23  
p6\*3.222 ph25  
p6\*3.167 ph23  
p6\*0.333 ph25  
p6\*2.722 ph23  
p6\*4.167 ph25  
p6\*2.944 ph23  
p6\*4.111 ph25

p6\*3.556 ph23  
p6\*4.556 ph25  
p6\*3.222 ph23  
p6\*3.167 ph25  
p6\*0.333 ph23  
p6\*2.722 ph25  
p6\*4.167 ph23  
p6\*2.944 ph25  
p6\*4.111 ph23  
lo to 24 times 13

;end DIPSI2

d21 p11:f1  
p1 ph19  
DELTA6

p2 ph17

4u  
p16:gp4 ;;gpz4 = 30  
d16  
4u BLKGRAD  
go=2 ph30

;PROCESSING

d11 do:f2 wr #0 if #0 zd ;use AU: splitx to separate datasets

d11\*0.15 igrad EA  
d11\*0.15 igrad EA3 ;HMBC  
d11\*0.15 igrad EA4 ;HMBC  
d11\*0.15 ip5\*2

d11\*0.15 iu9

lo to 2 times 2

d11\*0.15 id0  
d11\*0.15 id10

d11\*0.15 ip3\*2  
d11\*0.15 ip6\*2  
d11\*0.15 ip16\*2  
d11\*0.15 ip28\*2  
d11\*0.15 ip15\*2  
d11\*0.15 ip30\*2  
d11\*0.15 ip31\*2

```

d11*0.15 ru9
lo to 2 times 14 ; 14 = td1/6 NBL = 3 !!!

exit

ph1=0
ph2=1

ph21=0

ph3=0 2 2 0

ph4=2 2 0 0 0 2 2
ph5=1 1 3 3 3 3 1 1

ph6=0
ph7=0
ph8=0
ph9=0

ph22=1 1 1 1 3 3 3 3

ph23=3 ;DIPSI-2
ph25=1

ph11=(8) 1 1 1 1 7 7 7 7 ;TIG-BIRD phases - ph11 and ph12
ph12=(8) 5 5 5 5 3 3 3 3

ph16=0 2 ;1H-1H TOCSY with DIPSI-2
ph17=0
ph18=0 0 2 2
ph19=2 2 0 0

ph26=0 ;HMBC
ph27=0
ph28=0 2 2 0
ph29=0 0 2 2

ph15=0 2
ph30=0 2
ph31=0 2

;p10 : 0W
;p11 : f1 channel - power level for pulse (default)
;p12 : f2 channel - power level for pulse (default)
;p112: f2 channel - power level for CPD/BB decoupling
;sp7: f2 channel - shaped pulse 180 degree
;p1 : f1 channel - 90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p24: f2 channel - 180 degree shaped pulse for refocussing
; = 2msec for Crp80comp.4
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D) [3 usec]
;d10 : incremented delay (2D) [3 usec]
;d16: delay for homospoil/gradient recovery

;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 2 * n
;ds: >= 16
;td1: number of experiments
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

```

```

; ~~~~~ WaveMaker Shapes ~~~~~
;USE: wvm -a command in TopSpin command line to generate CA-WURST adiabatic shape
pulses

;sp18:wvm:ad180Cawurst: cawurst-20(121 ppm, 1.92 ms; H2L)

;sp20:wvm:ad180Cawurstlh2: cawurst-20(121 ppm, 0.97 ms; L2H)

;for z-only gradients:

;use gradient files:
;gpnam: SMSQ10.100

```

### **Pulse sequence code of NORD {HSQC}-{TOCSY} experiment for Bruker spectrometers (*Avance III and NEO systems*)**

```

;nord_hsqc_tocsy
;avance-version - tested on NEO system, TopSpin 4.0.2

;This pulse sequence is part of István Timári, Tamás Milán Nagy, Katalin E. Kövér,
;Ole W. Sørensen, 'Synergy and sensitivity-balance in concatenating experiments
;in NO Relaxation Delay NMR (NORD)'

;June 17, 2021 tested by IT, TMN, KEK

;The pulse sequence has been coded for test purposes only and may contain errors.
;The functionality of the pulse sequence itself may differ depending on
;the hardware as well as the software used to execute it. Functionality
;on differing systems cannot be granted.
;Any use of this pulse sequence on a spectrometer is at your own risk.
;
;By using this pulse sequence, or any modification of it in any published material
;you agree to acknowledge the above-mentioned publication.

;NORD concatenation of TWO experiments: HSQC and TOCSY

;TIG-BIRD and INEPT elements include J-compensated adiabatic X-inversion pulse

;set FNmode - echo-antiecho and NBL=2

;PROCESSING:
;FIRST separate the two datasets using AU splitx
;the resulting datasets can be processed with xfb (set echo-antiecho, pk)
;the F1 chemical shift scale in TOCSY is corrected with au-program fixF1 prior to xfb.

;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delayek.incl>

;use: wvm -a in TopSpin command line to setup adiabatic pulse parameters

"cnst2=113"          ;for J-compensated adiabatic pulse set CNST2 = 113

"p2=p1*2"
"p4=p3*2"

```

```

"d4=1s/(cnst2*4)" ;INEPT delay
"d6=1s/(cnst2*2)" ;TIG-BIRD delay, CNST2 = 113 Hz with adiabatic 13C-pulse

"d11=50u"

"d20=10u" ;z-filter delays in TOCSY
"d21=10u"

"p11=p1/2" ;1H-45 for TIG-BIRD with v = 0

"d0=3u"

"in0=inf1/2"

"d10=3u"

"cnst10=2*dw/inf1" ;cnst10 is a scaling factor for SW(13C)/SW(1H) for TOCSY
"in10=inf1*cnst10"

"DELTA=p16+d16+p2+d0*2-4u"

"DELTA2=d4-4u" ;INEPT-delay
"d22 = DELTA2-p29"

"d25=p16+d16+4u-p1*0.78+de"

"d24 = 1.86m"

"DELTA3=d24-cnst17*p24/2-4u" ;CNST17 = 1

"DELTA4=d6-4u" ;TIG-BIRD delay
"d23 = DELTA4-p31"

"DELTA5=p16+d16+d10" ;1H-1H TOCSY
"DELTA6=p16+d16+8u"

"FACTOR3=(d7/(p6*115.112))/2" ;d7 = 70-80 ms for 1H-1H TOCSY transfer
"l3=FACTOR3*2"

"l4 = td1/4" ;NBL = 2, data from two experiments are stored in one file

"acqt0=0"

"p12 = (p1/90)*cnst12" ;CNST12 = small flip angle excitation pulse of 1H-1H TOCSY

"l9=1"

baseopt_echo

1 ze
  d11 p12:f2
2 d11 do:f2 st0 ;NBL = 2
  d11 p12:f2 p11:f1

  (p3 ph1):f2
  50u UNBLKGRAD
  p16:gp12 *2.77 ;purging gradient
  d16 ;gpz12 = 17.13%

4 (p11 ph11):f1 ;TIG-BIRD with v = 0, p11 = 45 degree 1H; ph11 = phase 45
  DELTA4 p10:f2
  4u

  (p2 ph1)
  (p31:sp18 ph6):f2 ;J-compensated adiabatic pulse during TIG-BIRD,
  4u

```

```

d23
(p1 ph12):f1          ;ph12 = phase 225 degree

DELTA2                ;INEPT delay
4u

(p2 ph1)
(p29:sp20 ph6):f2    ;J-compensated adiabatic pulse during INEPT
4u

d22 p12:f2

(p1 ph22)
p16:gp11             ;purging gradient gpz11 = 31-51
d16

(p3 ph3):f2
d0

(p2 ph7)

d0
p16:gp1*EA
d16 p10:f2
(p24:sp7 ph8:r):f2   ;composite CHIRP, 2 ms
4u
DELTA p12:f2

(center (p1 ph21) (p3 ph4):f2 )
4u
DELTA3 p10:f2
(center (p2 ph1) (p24:sp7 ph9:r):f2 ) ;composite CHIRP, 2 ms
4u
DELTA3 p12:f2
(center (p1 ph2) (p3 ph5):f2 )
4u
DELTA2 p10:f2

(p2 ph1)
(p29:sp20 ph1):f2    ;J-compensated adiabatic pulse during back INEPT transfer
4u
d22 p11:f1

(p1 ph1)

d25

(p2 ph1)

4u p112:f2
p16:gp2              ;gpz2 = 20 %
d16 BLKGRAD

goscnp ph31 cpd2:f2

22 d11
23 d11 do:f2

50u UNBLKGRAD

p16:gp13             ;purging gradient between two experiments
;gpz13 = 17.13-71%

d16 p11:f1
p12 ph16            ;1H-1H TOCSY with DIPSI-2 starts here
;SECOND experiment with small-flip angle excitation, p12

if "19 %2 == 1"

{

```

```

DELTA5                ;p12 should be optimized, ca. 10-30 degree 1H pulse
                      ;p12 = (p1/90)*cnst12", CNST12 = small flip angle excitation
p2 ph17
d10                   ;d10 incremented by in10 to get 1H-dimension for TOCSY in F1
p16:gp3               ;gpz3 = 30
d16

}
else
{
  d10                  ;;d10 incremented by in10 to get 1H-dimension for TOCSY in F1
  p16:gp3*EA          ;;gpz3 = 30
  d16
  p2 ph17
  DELTA5
}

p1 ph18
d20 pl10:f1

                                ;begin DIPSI2
24 p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23

p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 24 times 13

                                ;end DIPSI2

d21 pl1:f1
p1 ph19

DELTA6
p2 ph17
4u
p16:gp4               ;gpz4 = 30
d16
4u BLKGRAD

```

```

go=2 ph30

d11 do:f2 wr #0 if #0 zd          ;PROCESSING
                                ;use AU: splitx to separate datasets

d11*0.33 igrad EA
d11*0.33 ip5*2

d11*0.33 iu9

lo to 2 times 2

d11*0.15 id0
d11*0.15 id10

d11*0.15 ip3*2
d11*0.15 ip6*2
d11*0.15 ip16*2
d11*0.15 ip30*2
d11*0.15 ip31*2

d11*0.15 ru9

lo to 2 times 14      ; 14 = td1/4   NBL = 2

exit

ph1=0
ph2=1

ph21=0

ph3=0 2 2 0

ph4=2 2 0 0 0 0 2 2
ph5=1 1 3 3 3 3 1 1

ph6=0
ph7=0
ph8=0
ph9=0

ph22=1 1 1 1 3 3 3 3

ph23=3      ;DIPSI-2
ph25=1

ph11=(8) 1 1 1 1 7 7 7 7      ;TIG-BIRD phases - ph11 and ph12
ph12=(8) 5 5 5 5 3 3 3 3

ph16=0 2          ;1H-1H TOCSY with DIPSI-2
ph17=0
ph18=0 0 2 2
ph19=2 2 0 0

ph30=0 2
ph31=0 2

;p10 : 0W
;p11 : f1 channel - power level for pulse (default)
;p12 : f2 channel - power level for pulse (default)
;p112: f2 channel - power level for CPD/BB decoupling
;sp7: f2 channel - shaped pulse 180 degree
;p1 : f1 channel - 90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p24: f2 channel - 180 degree shaped pulse for refocussing
      ; = 2msec for Crp80comp.4
;p16: homospoil/gradient pulse

```

```
;d0 : incremented delay (2D) [3 usec]
;d10 : incremented delay (2D) [3 usec]
;d16: delay for homospoil/gradient recovery

;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 2 * n
;ds: >= 16
;td1: number of experiments
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

; ~~~~~ WaveMaker Shapes ~~~~~
;USE: wvm -a command in TopSpin command line to generate CA-WURST adiabatic shape
pulses

;sp18:wvm:ad180Ccawurst: cawurst-20(121 ppm, 1.92 ms; H2L)

;sp20:wvm:ad180Ccawurstlh2: cawurst-20(121 ppm, 0.97 ms; L2H)

;for z-only gradients:

;use gradient files:
;gpnam: SMSQ10.100
```